// File: /contracts/HashedTimelockERC20.sol



// SPDX-License-Identifier: Blockchain Commodities
pragma solidity ^0.8.0;

import "openzeppelin-solidity/contracts/token/ERC20/IERC20.sol";
import "openzeppelin-solidity/contracts/token/ERC20/utils/SafeERC20.sol";
/**
* Hashed Timelock Contracts (HTLCs) on Ethereum ERC20 tokens.
*
* This contract provides a way to create and keep HTLCs for ERC20 tokens.
*
* See HashedTimelock.sol for a contract that provides the same functions
* for the native ETH token.
*
* Protocol:
*
*  1) newContract(receiver, hashlock, timelock, tokenContract, amount) - a
*      sender calls this to create a new HTLC on a given token (tokenContract)
*       for a given amount. A 32 byte contract id is returned
*  2) withdraw(contractId, preimage) - once the receiver knows the preimage of
*      the hashlock hash they can claim the tokens with this function
*  3) refund() - after timelock has expired and if the receiver did not
*      withdraw the tokens the sender / creator of the HTLC can get their tokens
*      back with this function.
 */
contract HashedTimelockERC20 {

    using SafeERC20 for IERC20;

    // PARAMS
    uint32 public feeNumerator;
    uint32 public feeDenominator;
    address public feeReceiver;
    address public owner;
    bytes32 private emptyLockContractHash;

    /**
     * Setting initial values for the smart contract when it is deployed, all these values can be configured later
     */
    constructor(address initialOwner, address initialFeeReceiver, uint32 initialFeeNumerator, uint32 initialFeeDenominator) {
        require(initialOwner != address(0), "40207: INVALID OWNER");
        require(initialFeeDenominator != 0, "40209: INVALID DENOMINATOR");
        owner = initialOwner;
        feeReceiver = initialFeeReceiver;
        feeNumerator = initialFeeNumerator;
        feeDenominator = initialFeeDenominator;
        emptyLockContractHash = getEmptyContractHash();
    }

    struct LockContract {
        uint256 amount;
        bytes32 hashLock;
        bytes32 preimage;
        uint32 timeLock;
        address sender;
        address receiver;
        address tokenContract;
        bool collected;
    }

    mapping(bytes32 => LockContract) public contracts;

    event HTLCERC20New(
        bytes32 indexed contractId,
        bytes32 hashlock,
        uint32 timelock,
        address indexed sender,
        address indexed receiver,
        address tokenContract,
        uint256 amount
    );
    event HTLCERC20Withdraw(bytes32 indexed contractId);
    event HTLCERC20Refund(bytes32 indexed contractId);
    event HTLCERC20NewOwner(address owner);
    event HTLCERC20NewFeeReceiver(address feeReceiver);
    event HTLCFeeUpdated(uint32 feeNumerator, uint32 feeDenominator);

    modifier onlyOwner() {
        require(msg.sender == owner, "40301: FORBIDDEN");
        _;
    }

    modifier onlyFeeReceiver() {
        require(msg.sender == feeReceiver, "40302: FORBIDDEN");
        _;
    }

    modifier tokensTransferable(address token, address sender, uint256 amount) {
        require(amount > 0, "40201: FUNDS REQUIRED");
        require(
            IERC20(token).allowance(sender, address(this)) >= amount, "40203: FUNDS REFUSAL"
        );
        _;
    }
    modifier futureTimeLock(uint32 time) {
        // only requirement is the time lock time is after the last block time (now).
        // probably want something a bit further in the future then this.
        // but this is still a useful sanity check:
        require(time > block.timestamp, "40011: INVALID VALUE");
        _;
    }
    modifier contractExists(bytes32 contractId) {
        require(haveContract(contractId), "40400: NOT FOUND");
        _;
    }
    modifier hashLockMatches(bytes32 contractId, bytes32 x) {
        require(
            contracts[contractId].hashLock == keccak256(abi.encodePacked(x)), "40100: UNAUTHORIZED"
        );
        _;
    }
    modifier withdrawable(bytes32 contractId) {
        require(contracts[contractId].receiver == msg.sender, "40303: FORBIDDEN");
        require(!contracts[contractId].collected, "40900: CONFLICT");
        _;
    }
    modifier refundable(bytes32 _contractId) {
        require(contracts[_contractId].sender == msg.sender, "40304: FORBIDDEN");
        require(!contracts[_contractId].collected, "40900: CONFLICT");
        require(contracts[_contractId].timeLock <= block.timestamp, "40101: UNAUTHORIZED");
        _;
    }


    /**
     * Sender / Payer sets up a new hash time lock contract depositing the
     * funds and providing the reciever and terms.
     *
     * NOTE: _receiver must first call approve() on the token contract.
     *       See allowance check in tokensTransferable modifier.

     * @param receiver Receiver of the tokens.
     * @param hashLock A keccak256 hash hashlock.
     * @param timeLock UNIX epoch seconds time that the lock expires at.
     *                  Refunds can be made after this time.
     * @param tokenContract ERC20 Token contract address.
     * @param amount Amount of the token to lock up.
     * @return contractId Id of the new HTLC. This is needed for subsequent
     *                    calls.
     */
    function newContract(
        address receiver,
        bytes32 hashLock,
        uint32 timeLock,
        address tokenContract,
        uint256 amount
    )
    external
    tokensTransferable(tokenContract, msg.sender, amount)
    futureTimeLock(timeLock)
    returns (bytes32 contractId)
    {
        contractId = keccak256(
            abi.encodePacked(
                msg.sender,
                receiver,
                tokenContract,
                amount,
                hashLock,
                timeLock
            )
        );

        // Reject if a contract already exists with the same parameters. The
        // sender must change one of these parameters (ideally providing a
        // different _hashLock).
        if (haveContract(contractId))
            revert("40901: CONFLICT");

        contracts[contractId] = LockContract(
            amount,
            hashLock,
            0x0,
            timeLock,
            msg.sender,
            receiver,
            tokenContract,
            false
        );

        emit HTLCERC20New(
            contractId,
            hashLock,
            timeLock,
            msg.sender,
            receiver,
            tokenContract,
            amount
        );

        // This contract becomes the temporary owner of the tokens
        IERC20(tokenContract).safeTransferFrom(msg.sender, address(this), amount);

        return contractId;
    }

    /**
    * Called by the receiver once they know the preimage of the hashlock.
    * This will transfer ownership of the locked tokens to their address.
    *
    * @param contractId Id of the HTLC.
    * @param preimage keccak256(_preimage) should equal the contract hashlock.
    * @return bool true on success
     */
    function withdraw(bytes32 contractId, bytes32 preimage)
    external
    contractExists(contractId)
    hashLockMatches(contractId, preimage)
    withdrawable(contractId)
    returns (bool)
    {
        uint256 amount = contracts[contractId].amount;
        address tokenContract = contracts[contractId].tokenContract;
        address receiver = contracts[contractId].receiver;
        contracts[contractId].preimage = preimage;
        contracts[contractId].collected = true;
        delete contracts[contractId].amount;
        delete contracts[contractId].hashLock;
        delete contracts[contractId].timeLock;
        delete contracts[contractId].sender;
        delete contracts[contractId].receiver;
        delete contracts[contractId].tokenContract;
        emit HTLCERC20Withdraw(contractId);
        if (feeReceiver != address(0)) {
            uint fee = amount * feeNumerator / feeDenominator;
            uint withdrawalAmount = amount - fee;
            IERC20(tokenContract).safeTransfer(receiver, withdrawalAmount);
            IERC20(tokenContract).safeTransfer(feeReceiver, fee);
        } else {
            IERC20(tokenContract).safeTransfer(receiver, amount);
        }
        return true;
    }

    /**
     * Called by the sender if there was no withdraw AND the time lock has
     * expired. This will restore ownership of the tokens to the sender.
     *
     * @param contractId Id of HTLC to refund from.
     * @return bool true on success
     */
    function refund(bytes32 contractId)
    external
    contractExists(contractId)
    refundable(contractId)
    returns (bool)
    {
        uint256 amount = contracts[contractId].amount;
        address tokenContract = contracts[contractId].tokenContract;
        address sender = contracts[contractId].sender;
        contracts[contractId].collected = true;
        delete contracts[contractId].amount;
        delete contracts[contractId].hashLock;
        delete contracts[contractId].timeLock;
        delete contracts[contractId].sender;
        delete contracts[contractId].receiver;
        delete contracts[contractId].tokenContract;
        emit HTLCERC20Refund(contractId);
        IERC20(tokenContract).safeTransfer(sender, amount);
        return true;
    }

    /**
     * Get contract details.
     * @param contractId HTLC contract id
     */
    function getContract(bytes32 contractId)
    external
    view
    returns (
        uint256 amount,
        bytes32 hashlock,
        bytes32 preimage,
        uint32 timelock,
        address sender,
        address receiver,
        address tokenContract,
        bool collected
    )
    {
        if (!haveContract(contractId))
            return (0, 0, 0, 0, address(0), address(0), address(0), false);
        LockContract memory fetchedContract = contracts[contractId];
        return (
        fetchedContract.amount,
        fetchedContract.hashLock,
        fetchedContract.preimage,
        fetchedContract.timeLock,
        fetchedContract.sender,
        fetchedContract.receiver,
        fetchedContract.tokenContract,
        fetchedContract.collected
        );
    }

    /**
     * Is there a contract with id _contractId.
     * @param contractId Id into contracts mapping.
     */
    function haveContract(bytes32 contractId)
    internal
    view
    returns (bool exists)
    {
        return keccak256(abi.encodePacked(contracts[contractId].amount,
            contracts[contractId].hashLock,
            contracts[contractId].preimage,
            contracts[contractId].timeLock,
            contracts[contractId].sender,
            contracts[contractId].receiver,
            contracts[contractId].tokenContract,
            contracts[contractId].collected))
        != emptyLockContractHash;
    }

    /**
     * This method return a keccak256 hash of an empty Lock contract.
     */
    function getEmptyContractHash() pure private returns (bytes32) {
        uint256 u256 = 0;
        bytes32 b32 = 0;
        uint32 u32 = 0;
        address add = address(0);
        bool b = false;
        return keccak256(abi.encodePacked(u256, b32, b32, u32, add, add, add, b));
    }


    /**
     * This method will update the feeTo address who can withdraw the
     * fee collected on this smart contract
     * @param feeReceiverAddress - address of new fee collector
     */
    function updateFeeReceiver(address feeReceiverAddress) onlyOwner external {
        feeReceiver = feeReceiverAddress;
        emit HTLCERC20NewFeeReceiver(feeReceiver);
    }

    /**
     * This method will update the owner address on this smart contract
     * @param ownerAddress - address of new owner
     */
    function updateOwner(address ownerAddress) onlyOwner external {
        require(ownerAddress != address(0), "40207: INVALID OWNER");
        owner = ownerAddress;
        emit HTLCERC20NewOwner(owner);
    }

    /**
     * This method will allow owner to update the fee charged upon successful withdrawal
     * @param updatedFeeNumerator - new fee numerator
     * @param updatedFeeDenominator - new fee denominator
     */
    function updateFee(uint32 updatedFeeNumerator, uint32 updatedFeeDenominator) onlyOwner external {
        require(updatedFeeNumerator != 0, "40208: INVALID NUMERATOR");
        require(updatedFeeDenominator != 0, "40209: INVALID DENOMINATOR");
        feeNumerator = updatedFeeNumerator;
        feeDenominator = updatedFeeDenominator;
        emit HTLCFeeUpdated(feeNumerator, feeDenominator);
    }

}

// File: openzeppelin-solidity/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: openzeppelin-solidity/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: openzeppelin-solidity/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}